

NAG Fortran Library Routine Document

G13CGF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

For a bivariate time series, G13CGF calculates the noise spectrum together with multiplying factors for the bounds and the impulse response function and its standard error, from the univariate and bivariate spectra.

2 Specification

```

SUBROUTINE G13CGF(XG, YG, XYRG, XYIG, NG, STATS, L, N, ER, ERLW, ERUP,
1              RF, RFSE, IFAIL)
      INTEGER      NG, L, N, IFAIL
real            XG(NG), YG(NG), XYRG(NG), XYIG(NG), STATS(4), ER(NG),
1              ERLW, ERUP, RF(L), RFSE

```

3 Description

An estimate of the noise spectrum in the dependence of series y on series x at frequency ω is given by

$$f_{y|x}(\omega) = f_{yy}(\omega)(1 - W(\omega)),$$

where $W(\omega)$ is the squared coherency described in G13CEF and $f_{yy}(\omega)$ is the univariate spectrum estimate for series y . Confidence limits on the true spectrum are obtained using multipliers as described for G13CAF, but based on $(d - 2)$ degrees of freedom.

If the dependence of y_t on x_t can be assumed to be represented in the time domain by the one sided relationship

$$y_t = v_0 x_t + v_1 x_{t-1} + \dots + n_t,$$

where the noise n_t is independent of x_t , then it is the spectrum of this noise which is estimated by $f_{y|x}(\omega)$.

Estimates of the impulse response function $v_0, v_1, v_2, \dots$ may also be obtained as

$$v_k = \frac{1}{\pi} \int_0^\pi \operatorname{Re} \left(\frac{\exp(ik\omega) f_{xy}(\omega)}{f_{xx}(\omega)} \right) d\omega,$$

where Re indicates the real part of the expression. For this purpose it is essential that the univariate spectrum for x , $f_{xx}(\omega)$, and the cross spectrum, $f_{xy}(\omega)$, be supplied to this routine for a frequency range

$$\omega_l = \left[\frac{2\pi l}{L} \right], \quad 0 \leq l \leq [L/2],$$

where $[]$ denotes the integer part, the integral being approximated by a finite Fourier transform.

An approximate standard error is calculated for the estimates v_k . Significant values of v_k in the locations described as anticipatory responses in the parameter array RF indicate that feedback exists from y_t to x_t . This will bias the estimates of v_k in any causal dependence of y_t on $x_t, x_{t-1}, \dots$

4 References

Jenkins G M and Watts D G (1968) *Spectral Analysis and its Applications* Holden-Day

Bloomfield P (1976) *Fourier Analysis of Time Series: An Introduction* Wiley

5 Parameters

- 1: XG(NG) – *real* array *Input*
On entry: the NG univariate spectral estimates, $f_{xx}(\omega)$, for the x series.

- 2: YG(NG) – *real* array *Input*
On entry: the NG univariate spectral estimates, $f_{yy}(\omega)$, for the y series.

- 3: XYRG(NG) – *real* array *Input*
On entry: the real parts, $cf(\omega)$, of the NG bivariate spectral estimates for the x and y series. The x series leads the y series.

- 4: XYIG(NG) – *real* array *Input*
On entry: the imaginary parts, $qf(\omega)$, of the NG bivariate spectral estimates for the x and y series. The x series leads the y series.

Note: the two univariate and bivariate spectra must each have been calculated using the same method of smoothing. For rectangular, Bartlett, Tukey or Parzen smoothing windows, the same cut-off point of lag window and the same frequency division of the spectral estimates must be used. For the trapezium frequency smoothing window, the frequency width and the shape of the window and the frequency division of the spectral estimates must be the same. The spectral estimates and statistics must also be unlogged.

- 5: NG – INTEGER *Input*
On entry: the number of spectral estimates in each of the arrays XG, YG, XYRG, XYIG. It is also the number of noise spectral estimates.
Constraint: $NG \geq 1$.

- 6: STATS(4) – *real* array *Input*
On entry: the four associated statistics for the univariate spectral estimates for the x and y series. STATS(1) contains the degree of freedom, STATS(2) and STATS(3) contain the lower and upper bound multiplying factors respectively and STATS(4) contains the bandwidth.
Constraints:

$$\begin{aligned} \text{STATS}(1) &\geq 3.0, \\ 0.0 < \text{STATS}(2) &\leq 1.0, \\ \text{STATS}(3) &\geq 1.0. \end{aligned}$$

- 7: L – INTEGER *Input*
On entry: the frequency division, L , of the spectral estimates as $\frac{2\pi}{L}$. It is also the order of the FFT used to calculate the impulse response function. L must relate to the parameter NG by the relationship.
Constraints:

$$NG = [L/2] + 1.$$

The largest prime factor of L must not exceed 19, and the total number of prime factors of L, counting repetitions, must not exceed 20. These two restrictions are imposed by C06EBF which performs the FFT.

- 8: N – INTEGER *Input*
On entry: the number of points in each of the time series x and y . N should have the same value as NXY in the call of G13CCF or G13CDF which calculated the smoothed sample cross spectrum. N is used in calculating the impulse response function standard error (RFSE).
Constraint: $N \geq 1$.

- 9: ER(NG) – *real* array *Output*
On exit: the NG estimates of the noise spectrum, $\hat{f}_{y|x}(\omega)$ at each frequency.
- 10: ERLW – *real* *Output*
On exit: the noise spectrum lower limit multiplying factor.
- 11: ERUP – *real* *Output*
On exit: the noise spectrum upper limit multiplying factor.
- 12: RF(L) – *real* array *Output*
On exit: the impulse response function. Causal responses are stored in ascending frequency in RF(1) to RF(NG) and anticipatory responses are stored in descending frequency in RF(NG + 1) to RF(L).
- 13: RFSE – *real* *Output*
On exit: the impulse response function standard error.
- 14: IFAIL – INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. Users who are unfamiliar with this parameter should refer to Chapter P01 for details.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).
 For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, because for this routine the values of the output parameters may be useful even if IFAIL $\neq$ 0 on exit, the recommended value is -1. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, NG < 1,
 or STATS(1) < 3.0,
 or STATS(2) $\leq$ 0.0,
 or STATS(2) > 1.0,
 or STATS(3) < 1.0,
 or N < 1.

IFAIL = 2

A bivariate spectral estimate is zero. For this frequency the noise spectrum is set to zero, and the contribution to the impulse response function and its standard error is set to zero.

IFAIL = 3

A univariate spectral estimate is negative. For this frequency the noise spectrum is set to zero, and the contributions to the impulse response function and its standard error are set to zero.

IFAIL = 4

A univariate spectral estimate is zero. For this frequency the noise spectrum is set to zero and the contributions to the impulse response function and its standard error are set to zero.

IFAIL = 5

A calculated value of the squared coherency exceeds 1.0. For this frequency the squared coherency is reset to 1.0 with the consequence that the noise spectrum is zero and the contribution to the impulse response function at this frequency is zero.

IFAIL = 6

On entry, $[L/2] + 1 \neq NG$,
 or L has a prime factor exceeding 19,
 or L has more than 20 prime factors, counting repetitions.

If more than one failure of types 2, 3, 4 and 5 occurs then the failure type which occurred at lowest frequency is returned in IFAIL. However the actions indicated above are also carried out for failures at higher frequencies.

7 Accuracy

The computation of the noise is stable and yields good accuracy. The FFT is a numerically stable process, and any errors introduced during the computation will normally be insignificant compared with uncertainty in the data.

8 Further Comments

The time taken by the routine is approximately proportional to NG.

9 Example

The example program reads the set of univariate spectrum statistics, the two univariate spectra and the cross spectrum at a frequency division of $\frac{2\pi}{20}$ for a pair of time series. It calls G13CGF to calculate the noise spectrum and its confidence limits multiplying factors, the impulse response function and its standard error. It then prints the results.

9.1 Program Text

Note: the listing of the example program presented below uses *bold italicised* terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```
*      G13CGF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NGMAX, LMAX
      PARAMETER        (NGMAX=9,LMAX=16)
      INTEGER          NIN, NOUT
      PARAMETER        (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real             ERLW, ERUP, RFSE
      INTEGER          I, IFAIL, J, L, N, NG
*      .. Local Arrays ..
      real             ER(NGMAX), RF(LMAX), STATS(4), XG(NGMAX),
+                    XYIG(NGMAX), XYRG(NGMAX), YG(NGMAX)
*      .. External Subroutines ..
      EXTERNAL         G13CGF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G13CGF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) NG, L, N
      IF (NG.GT.0 .AND. NG.LE.NGMAX .AND. L.GT.0 .AND. L.LE.LMAX) THEN
        READ (NIN,*) (STATS(I),I=1,4)
        READ (NIN,*) (XG(I),YG(I),XYRG(I),XYIG(I),I=1,NG)
        IFAIL = 1
      
```

```

      CALL G13CGF(XG,YG,XYRG,XYIG,NG,STATS,L,N,ER,ERLW,ERUP,RF,RFSE,
+           IFAIL)
*
      WRITE (NOUT,*)
      IF (IFAIL.NE.0) THEN
        WRITE (NOUT,99999) 'G13CGF fails. IFAIL =', IFAIL
        WRITE (NOUT,*)
      END IF
      IF (IFAIL.NE.1) THEN
        WRITE (NOUT,*) '          Noise spectrum'
        DO 20 J = 1, NG
          WRITE (NOUT,99998) J - 1, ER(J)
20      CONTINUE
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Noise spectrum bounds multiplying factors'
        WRITE (NOUT,99997) 'Lower =', ERLW, '          Upper =', ERUP
        WRITE (NOUT,*)
        WRITE (NOUT,*) 'Impulse response function'
        WRITE (NOUT,*)
        DO 40 J = 1, L
          WRITE (NOUT,99998) J - 1, RF(J)
40      CONTINUE
        WRITE (NOUT,*)
        WRITE (NOUT,99997)
+           'Impulse response function standard error =', RFSE
      END IF
      END IF
      STOP
*
99999 FORMAT (1X,A,I3)
99998 FORMAT (1X,I5,F16.4)
99997 FORMAT (1X,A,F10.4,A,F10.4)
      END

```

9.2 Program Data

G13CGF Example Program Data

9	16	296		
30.00000	.63858	1.78670	.33288	
2.03490	21.97712	-6.54995	0.00000	
.51554	3.29761	.34107	-1.19030	
.07640	.28782	.12335	.04087	
.01068	.02480	-.00514	.00842	
.00093	.00285	-.00033	.00032	
.00100	.00203	-.00039	-.00001	
.00076	.00125	-.00026	.00018	
.00037	.00107	.00011	-.00016	
.00021	.00191	.00007	0.00000	

9.3 Program Results

G13CGF Example Program Results

	Noise spectrum
0	0.8941
1	0.3238
2	0.0668
3	0.0157
4	0.0026
5	0.0019
6	0.0011
7	0.0010
8	0.0019

Noise spectrum bounds multiplying factors
 Lower = 0.6298 Upper = 1.8291

Impulse response function

0	-0.0547
---	---------

1	0.0586
2	-0.0322
3	-0.6956
4	-0.7181
5	-0.8019
6	-0.4303
7	-0.2392
8	-0.0766
9	0.0657
10	-0.1652
11	-0.0439
12	-0.0494
13	-0.0384
14	0.0838
15	-0.0814

Impulse response function standard error = 0.0863
